

CSU22022 Computer Architecture I

Third Lecture

Michael Manzke

2023-2024

Trinity College Dublin

Concurrent Signal Assignment Statement (CSA)

- Digital systems operate with concurrent signals
- Signals are assigned values at a specific point in time
- VHDL uses signal assignment statements
 - To specify value and time
- Multiple signal assignment statements are executed concurrently
 - The interface
 - Concurrent Signal Assignment Statement (CSA)

Half-Adder CSA

```
1 architecture Behavioral of HalfAdder is
2
3 begin
4
5     sum <= x xor y after 5ns;
6     carry <= x and y after 5ns;
7
8 end Behavioral;
```

- VHDL must specify:
 - Events
 - Delays
 - Concurrency

- Behavioral
 - Name of architecture that defines the HalfAdder entity
- Signal assignment statements
 - `sum <= x xor y after 5ns;`
 - `carry <= x and y after 5ns;`
- "`<=`" Signal assignment operator
 - Describes how output signals depend on input signals
- An output signal changes if an input signal has changed

Half Adder Operation




Figure 1: Input and output signals over time [ns]

- Signal propagation of the xor and and gates must be taken into account
 - Both gates require 5 ns propagation delay
- Signal assignment statements define this through the after Keyword
- This keyword specifies when the output signal is set to the result of an evaluation after an input signal transition
 - Event
- The textual order of the assignment statement has no influence on the timing

library and use clauses

```
1 library IEEE;  
2 use IEEE.STD_LOGIC_1164.ALL;  
3  
4 entity HalfAdder is  
5     .  
6     .  
7     .
```

- A library contains design entities that can be used
- The **library** IEEE clause defines the IEEE library
- The library may contain packages
- The above example specifies through the **use** clause the IEEE.STD_LOGIC_1164.**ALL** packages
- This package is required for std_logic type declaration

Full-Adder

```
1 library IEEE;
2 use IEEE.STD_LOGIC_1164.ALL;
3
4 entity FullAdder is
5     Port (in1,in2,c_in : in std_ulogic;
6           sum, c_out: out std_ulogic);
7 end FullAdder;
8
9 architecture Behavioral of FullAdder is
10     signal s1, s2, s3 : std_ulogic;
11     constant gate_delay : Time := 5ns;
12 begin
13     s1 <= in1 xor in2 after gate_delay;
14     s2 <= c_in and s1 after gate_delay;
15     s3 <= in1 and in2 after gate_delay;
16     sum <= s1 xor c_in after gate_delay;
17     c_out <= s2 or s3 after gate_delay;
18 end Behavioral;
```

Listing 1: With signal s1 s2 s3

Full-Adder Schematic




Figure 2: Schematic with signal `s1` `s2` `s3`

- `s1` <= `in1` xor `in2` after gate_delay;
- `s2` <= `c_in` and `s1` after gate_delay;
- `s3` <= `in1` and `in2` after gate_delay;
- `sum` <= `s1` xor `c_in` after gate_delay;
- `c_out` <= `s2` or `s3` after gate_delay;

Declaration and Body

```
1 architecture Behavioral of FullAdder is
2     signal s1, s2, s3 : std_ulogic;
3     constant gate_delay : Time := 5ns;
4 begin
```

Listing 2: Architecture Declaration

```
1 begin
2     s1 <= in1 xor in2 after gate_delay;
3     s2 <= c_in and s1 after gate_delay;
4     s3 <= in1 and in2 after gate_delay;
5     sum <= s1 xor c_in after gate_delay;
6     c_out <= s2 or s3 after gate_delay;
7 end Behavioral;
```

Listing 3: Architecture Body

The Full-Adder Model

- The full-adder simulates the signal transitions at gate-level
- The model has three internal signals
- These signals are not ports to the entity
- The internal signals are declared in the architectural declaration
- The Boolean equations define how each signal is derived as function of:
 - Other signals
 - Propagation delay
- **constant** can be used to declare a constant of a particular type
 - In this case **Time**

- A **signal** is not a variables
 - History of values over time
 - Waveform

```
1  signal s1, s2, s3 : std_ulogic := '0';
```

- Signals may use the assignment symbol `:=` followed by an expression
- The value of the expression will be initial value of the signal
- If no initialisation is provided the signal receives a default value
- VHDL signal types:
 - Integer, real, bit_vector...

```
1 sum <= s1 xor c_in after 5ns;
```

- In a more general form:
 - `signal <= value expression after time expression;`
- In the example:
 - if s1 or c_in change its value the sum will be assigned the result of the S1 xor c_in evaluation after 5ns
- The Time-value pair represents the future value of the signal
 - Also called transaction

Multiple signal Transactions

```
1  s1 <= x xor y after 5 ns, x or y after 10 ns, not x  
    after 10 ns;
```

Listing 4: It is possible to specify multiple signal transactions

- After one of the signals changed all three waveform elements will be evaluated and scheduled according to their after specification
- The simulation keeps an ordered list of all transactions scheduled for a particular signal scheduled transactions are also known as:
 - Projected output waveform